

Name: _____

Class: D16

Roll No: _____

EXPERIMENT NO: 05

Aim: To receive the value of temperature and the humidity of the DHT-11 sensor obtained on mobile using BLYNK APP along with the addition of 2 Leds.

Apparatus: Blynk App, Node MCU (ESP8266), DHT11, breadboard, connecting wires, LED

Prerequisite: Blynk App on mobile phone and create your login, library for Blynk in Arduino IDE

Theory:

Blynk can control Digital and Analog I/O Pins on your hardware directly. You don't even need to write code for it. Anything you connect to your hardware will be able to talk to Blynk. With Virtual Pins you can send something from the App. process it on a microcontroller and then send it back to the smartphone.

BLYNK APP

Blynk was designed for the Internet of Things It can control hardware remotely, it can display sensor data, it can store data, visualize it and do many other cool things

There are three major components in the platform

Blynk App - allows the user to create amazing interfaces for the projects using various widgets provided **Blynk Server** - responsible for all the communications between the smartphone and hardware. One can use the Blynk Cloud or run the private Blynk server locally

Blynk Libraries - for all the popular hardware platforms - enable communication with the server and process all the incoming and outgoing commands

FEATURES:

- A. Similar API & UI for all supported hardware & devices
- B. Connection to the cloud using

- C. WiFi
- D. Bluetooth and BLE
- E. Ethernet
- F. USB (Serial)
- G. GSM
- H. Set of easy-to-use Widgets.
- I. Direct pin manipulation with no code writing
- J. Easy to integrate and add new functionality using virtual pins.
- K. History data monitoring via SuperChart widget.
- L. Device-to-Device communication using Bridge Widget.
- M. Sending emails, tweets, push notifications, etc.
- N. New features can be constantly added!

PROCEDURE:

1. Create a New Project in the BLYNK app.
2. Write Project name Temperature Humidity and Select NodeMCU from drop down.
3. An AUTH token will be sent to your registered email, note this down.
4. Tap on the screen and add 2 Gauges.
5. Tap on the Widget arid select the respective Virtual pins for temperature and humidity data (V0 for temperature and V1 for humidity).
6. Add two LED pins.

CODE:

```
#define BLYNK_TEMPLATE_ID "TMPL6EQyNZGmf"
#define BLYNK_TEMPLATE_NAME "Weather Monitor"
#define BLYNK_AUTH_TOKEN "csO29wsYOwiV9aGX311ifPXNF23CTmy-"

#define BLYNK_PRINT Serial
#include <ESP8266WiFi.h>
#include <BlynkSimpleEsp8266.h>
#include <DHT.h>

char auth[] = BLYNK_AUTH_TOKEN;

char ssid[] = "shreyash"; // Your WiFi SSID
char pass[] = "shreyash"; // Your WiFi Password

BlynkTimer timer;

#define DHTPIN D7      // DHT sensor pin
#define DHTTYPE DHT11
DHT dht(DHTPIN, DHTTYPE);

// ===== LED Pins =====
#define LED1 D5 // Connect LED1 here (with resistor ~220Ω)
#define LED2 D6 // Connect LED2 here

// ===== Blynk Virtual Pins =====
#define VPIN_LED1 V2 // Virtual button in Blynk app
#define VPIN_LED2 V3 // Virtual button in Blynk app

void sendSensor() {
    float h = dht.readHumidity();
    float t = dht.readTemperature(); // Celsius

    if (isnan(h) || isnan(t)) {
        Serial.println("Failed to read from DHT sensor!");
        return;
    }

    // Send sensor values to Blynk
    Blynk.virtualWrite(V0, t);
    Blynk.virtualWrite(V1, h);

    Serial.print("Temperature : ");
    Serial.print(t);
    Serial.print(" °C  Humidity : ");
    Serial.print(h);
    Serial.println(" %");
}
```

```
// ===== Blynk Button Handlers =====
BLYNK_WRITE(VPIN_LED1) {
    int value = param.asInt(); // Get value from Blynk (0/1)
    digitalWrite(LED1, value); // Write to LED1
    Serial.print("LED1 is ");
    Serial.println(value ? "ON" : "OFF");
}

BLYNK_WRITE(VPIN_LED2) {
    int value = param.asInt();
    digitalWrite(LED2, value);
    Serial.print("LED2 is ");
    Serial.println(value ? "ON" : "OFF");
}

void setup() {
    Serial.begin(115200);


    Blynk.begin(auth, ssid, pass);
    dht.begin();

    // Setup LEDs as OUTPUT
    pinMode(LED1, OUTPUT);
    pinMode(LED2, OUTPUT);
    digitalWrite(LED1, LOW);
    digitalWrite(LED2, LOW);

    // Send sensor data every 1 second
    timer.setInterval(1000L, sendSensor);
}

void loop() {
    Blynk.run();
    timer.run();
}
```

OUTPUT:



CONCLUSION:

1. Learned how to interface DHT11 with NODEMCU ESP8266.
2. Accessed ESP8266 from Blynk App.
3. Learned to create a dashboard on Blynk in order to display data from DHT11.
4. Created conditions for LEDs of display to glow when temperature/humidity is not within limits.
5. Platforms like blynk may open up possibilities for applications like home automation.